set hive.exec.dynamic.partition = true;
set hive.exec.dynamic.partition.mode = 'nonstrict';
set hive.exec.max.dynamic.partitions = 200;
set hive.exec.max.dynamic.partitions.pernode = 200;
insert overwrite table jms_dm.dm_tab_loopbag_buildbag_summary_dt partition (dt)
select network_type                                                  --网点类型
     , proxy_area_code                                               --建包对应网点代理区编码
     , region_code                                                   --建包对应网点大区编码
     , transfer_type_desc                                            --建包类型
     , package_source                                                --建包来源
     , site_code                                                     --网点编码
     , site_name                                                     --网点名称
     , region_name                                                   --建包对应网点大区名称
     , proxy_area_name                                               --建包对应网点代理区名称
     , sum(package_qty)                               as exact_sum   --包内件
     , count(package_code)                            as bagging_qty --包数量
     , sum(if(is_break = 1 and exact_flag = 1, 1, 0)) as getout_qty  --循环袋流向违规数量
     , sum(if(exact_flag = 1, 1, 0))                  as recyle_qty  -- 循环袋建包数
     , sum(if(exact_flag = 0, 1, 0))                  as exact_qty   -- 非循环袋建包数
     , sum(if(exact_flag = 1, package_qty, 0))        as recyle_sum  --循环袋包内件
     , day_time                                       as scan_time   --日器
     , good_type                                                     --物品类型1是普通类型 2是文件类型
     , day_time                                       as dt
from jms_dm.dm_tab_loopbag_buildbag_detail_dt
where dt = date_add('{{ execution_date | cst_ds }}', -6)
   or dt = date_add('{{ execution_date | cst_ds }}', -2)
   or dt = '{{ execution_date | cst_ds }}'
group by network_type       --网点类型
       , proxy_area_code    --建包对应网点代理区编码
       , region_code        --建包对应网点大区编码
       , transfer_type_desc --建包类型
       , package_source     --建包来源
       , site_code          --网点编码
       , site_name          --网点名称
       , region_name        --建包对应网点大区名称
       , proxy_area_name    --建包对应网点代理区名称
       , day_time
       , good_type          --物品类型1是普通类型 2是文件类型
distribute by dt
;
